SAS® Code and Macros: How They Interact
Bruce Gilsen, Federal Reserve Board

INTRODUCTION

SAS users at every level of experience seem to struggle to understand the interaction between SAS code (DATA and PROC steps) and macros. Understanding this interaction enables SAS users to understand existing applications and to take better advantage of the tools offered by SAS software and develop code much more effectively. This important concept is first reviewed in a somewhat informal way, similar to the one-on-one discussions I have had many times with Federal Reserve Board SAS users, and then with a series of simple examples.

Macros Generate Code

The following explanation, which is informal and over-simplified, echoes many one-on-one discussions I have had with Federal Reserve Board SAS users. It is intended to provide an intuitive understanding of macro and SAS code processing. For more information, see the examples below. For a more formal and detailed explanation, see the chapters SAS Programs and Macro Processing and Macro Processing in the SAS 9.1.3 Macro Language Reference.

Over 20 years ago, a long-departed SAS expert explained macro execution with a simple mantra: "Macros generate code." I still consider this an extremely helpful way to start thinking about macro execution. I think of macros as containing two things.

- Macro language elements (%DO, %IF, %PUT, %STR, etc.)
- SAS (non-macro) code which the macro processor treats as literal text

During macro execution, the macro processor executes macro language elements and generates zero or more lines of SAS code. When macro processing completes, or at any step boundary, if any lines of SAS code were generated, they are processed as if the user typed them at that location.

Understanding the interaction of SAS (DATA and PROC step) code and macros enables users to understand existing applications and easily code applications that might otherwise be much more difficult. For example, macros can be used to easily generate SAS code that can be

- a large block of code such as a DATA or PROC step or multiple steps
- a single SAS statement
- part of a SAS statement

Let’s consider some examples. Examples 1 - 3 are fairly simple. Examples 4 and 5 are slightly less simple, but show an important coding concept: how to iteratively generate SAS code in a macro. Example 6 is more complex and illustrates a more automated application.

In the examples below, SAS code being “generated” means that the SAS code is treated as constant text by the macro processor, and is placed on the input stack to be processed when the macro finishes executing or at a step boundary, just as if the user had typed the code at that location in the program.

Example 1. SAS code but no macro language elements.

Macro MAC1 contains a DATA step but no macro language elements. While not a very interesting macro application in real life, this macro illustrates at the simplest level the relationship between SAS code and macros.

```
%macro mac1;                      (1)
data one;                         (2)
x1=11;                           (3)
x2=22;                           (4)
x3=33;                           (5)
run;                              (6)
%mend mac1;                       (7)
%mac1                              (8)
```

(1) - (7) define macro MAC1. After the macro processor processes (7), MAC1 is compiled. (8) invokes MAC1, which executes
as follows.

- (2) - (6) are SAS code and are generated, just as if the user had typed them at that location in the program. The following SAS code is generated.

```sas
data one;
  x1=11;
  x2=22;
  x3=33;
run;
```

- (6) is SAS code that specifies a step boundary. It causes DATA step ONE to be compiled and executed.

- (7) is a %MEND statement that ends macro execution.

A macro of this type could be useful if the same DATA step was needed in more than one location in an application. You could just invoke the macro with %MAC1 wherever the DATA step was needed.

### Example 2. SAS code and macro language elements with a step boundary.

Macro MAC2 contains a DATA step and some %LET and %PUT statements, which are macro language elements. As noted above, any SAS code generated by a macro is processed (compiled and executed) when macro processing completes or at any step boundary. This example illustrates the effect of step boundaries.

```sas
%macro mac2;
  %local blah;
  data one;
    x=1;
  %let blah=11;
  %put in DATA step before PUT statement blah=&blah;
  put x=;
  %let blah=22;
  %put in DATA step after PUT statement blah=&blah;
  run;
%put after DATA step;
%mend mac2;
%mac2
```

Macro MAC2 executes as follows. Text written to the SAS log when MAC2 executes is included at the end of this section.

- (1) is a macro language element and is immediately executed. Macro variable BLAH is created and is local to this macro.

- (2) and (3) are SAS code and are generated.

- (4) is a macro language element and is immediately executed. Macro variable BLAH is assigned the value 11.

- (5) is a macro language element and is immediately executed. The text in this statement is written to the SAS log.

- (6) is SAS code and is generated.

- (7) is a macro language element and is immediately executed. Macro variable BLAH is assigned the value 11.

- (8) is a macro language element and is immediately executed. The text in this statement is written to the SAS log.

- (9) is SAS code that specifies a step boundary. DATA step ONE comprising SAS statements (2), (3), (6), and (9) is compiled and executed. The PUT statement, (6), writes x=1 to the SAS log. Note in the SAS log shown below that (5) and (8) are written to the SAS log before (6).

- (10) is a macro language element and is immediately executed. The text in this statement is written to the SAS log.
(11) is a %MEND statement that ends macro execution.

The following is written to the SAS log when MAC2 executes.

```
in DATA step before PUT statement blah=11                                       <==== from (5)
in DATA step after PUT statement blah=22                                         <==== from (8)
x=1                                                                      <==== from (6)
NOTE: The data set WORK.ONE has 1 observations and 1 variables.                <==== from (6)
NOTE: DATA statement used (Total process time):
       real time           0.00 seconds
       cpu time            0.01 seconds
after DATA step                                                             <==== from (10)
```

Example 3. Use macro logic to conditionally generate a SAS statement.

In this example, we want a DATA step to include the statement X2=22; only when the value of macro variable INCLUDEX2 is 1.

DATA step when INCLUDEX2 is 1:
```
data one;
x1=11;
x2=22;
x3=33;
run;
```

DATA step when INCLUDEX2 is not 1:
```
data one;
x1=11;
x3=33;
run;
```

Let’s review the execution of macro MAC3 to see how it differs when INCLUDEX2 is or is not equal to 1. In a real application, INCLUDEX2’s value could be generated within the application, but for simplicity, we just assign its value with a %LET statement.

Example 3.a. When INCLUDEX2 is equal to 1.
```
%let includex2=1;
%macro mac3(macvar=);               (1)
data one;                         (2)
x1=11;                           (3)
%if &macvar=1 %then %do;        (4)
x2=22;                          (5)
%end;                           (6)
x3=33;                           (7)
run;                             (8)
%mend mac3;                         (9)
%mac3(macvar=&includex2)            (10)
```

(10) invokes macro MAC3. Macro variable MACVAR, which is local to this macro, has the value 1. MAC3 executes as follows.

- (2) and (3) are SAS code and are generated.
- In (4), the macro processor replaces the macro variable MACVAR with its value, 1.
  ```
  %if &macvar=1 %then %do;               (original statement)
  %if 1=1 %then %do;                    (after macro variable replaced with its value)
  ```

Foundation and Fundamentals

NESUG 2010
1=1 is true, so the macro processor processes the code between %DO and %END. (5) is SAS code and is generated.

- (7) and (8) are SAS code and are generated. (8) is a step boundary, so the following SAS statements are compiled and executed.

    data one;
    x1=11;
    x2=22;
    x3=33;
    run;

- (9) is a %MEND statement that ends macro execution.

Example 3.b. When INCLUDEX2 is not equal to 1.

%let includex2=-999;

%macro mac3(macvar=);               (1)
    data one;                        (2)
        x1=11;                        (3)
        %if &macvar=1 %then %do;        (4)
            x2=22;                    (5)
            %end;                           (6)
        x3=33;                        (7)
        run;                           (8)
%mend mac3;                         (9)

%mac3(macvar=&includex2)            (10)

Now, when MAC3 executes, macro variable MACVAR has the value -999. MAC3, repeated from above for convenience, executes as follows.

- (2) and (3) are SAS code and are generated.
- In (4), the macro processor replaces the macro variable MACVAR with its value, -999.

    %if -999=1 %then %do;                                (after macro variable replaced with its value)
    -999=1 is false, so the macro processor does not process the code between %DO and %END. (5) is not generated.
- (7) and (8) are SAS code and are generated. (8) is a step boundary, so the following SAS statements are compiled and executed.

    data one;
    x1=11;
    x3=33;
    run;

- (9) is a %MEND statement that ends macro execution.

Example 4. Use a macro loop to generate clauses in a SAS statement.

Data set ONE contains variables X1, X2, and X3, to be renamed to Y1, Y2, and Y3.

    data one;
    x1=11;
    x2=22;
    x3=33;
    run;
A DATA step to create data set TWO and rename the variables could be coded as follows.

```sas
data two;
  set one;
  rename
    x1=y1
    x2=y2
    x3=y3;
run;
```

The same DATA step code can be generated by a macro. This is more generalized, though in a real-world application, the upper bound of the %DO loop would probably be determined dynamically in the program, not hard-coded as 3.

```sas
%macro mac4;
  %local j; /* local macro variable */
  data two;
  set one;
  rename
    %do j = 1 %to 3;
      x&j = y&j
    %end;
run;
%mend mac4;
%mac4;
```

The RENAME statement generated by the macro has three components.

- "RENAME" precedes the macro loop. It is generated by (5).

- Each iteration of the macro loop, (6) - (8), generates one clause of the RENAME statement. Since each clause is part of a SAS statement, a semicolon must not be included as part of (7).

  - The first time, it generates: \( x_1 = y_1 \)
  - The second time, it generates: \( x_2 = y_2 \)
  - The third time, it generates: \( x_3 = y_3 \)

- A semicolon follows the macro loop. It is generated by (9).

Macro MAC4 executes as follows.

- (3) - (5) are SAS code and are generated.

- (6) - (8) are executed by the macro processor as follows.

  - (6) \( J \) is replaced with its value, 1. 1 is within the loop bounds, 1 to 3, so %DO loop processing continues.
  - (7) The following SAS code is generated: \( x_1 = y_1 \)
  - (8) %END ends the %DO statement, and the macro returns to (6).

  - (6) \( J \) is incremented from 1 to 2. 2 is within the loop bounds, 1 to 3, so %DO loop processing continues.
  - (7) The following SAS code is generated: \( x_2 = y_2 \)
  - (8) %END ends the %DO statement, and the macro returns to (6).

  - (6) \( J \) is incremented from 2 to 3. 3 is within the loop bounds, 1 to 3, so %DO loop processing continues.
  - (7) The following SAS code is generated: \( x_3 = y_3 \)
  - (8) %END ends the %DO statement, and the macro returns to (6).

  - (6) \( J \) is incremented from 3 to 4. 4 is outside the loop bounds, 1 to 3, so %DO loop processing ends, and processing resumes with (9).
(9) and (10) are SAS code and are generated. (9) is a semicolon that completes the RENAME statement. (10), a RUN statement, is a step boundary that ends the DATA step and causes the following SAS statements to be compiled and executed.

```sas
data two;
  set one;
  rename x1 = y1
     x2 = y2
     x3 = y3 ;
run;
```

(11) is a %MEND statement that ends macro execution.

This program shows that it is simple to iteratively generate part of a SAS statement in a macro once you understand the interaction between SAS code (DATA and PROC steps) and macros. This is an important technique with many possible uses, though this particular program is more easily coded with the following RENAME statement.

```sas
rename x1-x3 = y1-y3;
```

**Example 5. Use a macro loop to generate clauses in a SAS statement: an alternate approach.**

An equivalent way to code Example 4 is to include just the macro loop code in the macro, as follows.

```sas
%macro mac5;                      (1)
  %local j;                       (2)
  %do j = 1 %to 3;                (3)
    x&j = y&j                     (4)
  %end;                           (5)
%mend mac5;                       (6)

data two;                         (7)
  set one;                        (8)
  rename                          (9)
    %mac5                         (10)
  ;                               (11)
run;                              (12)
```

This program executes as follows.

- (1) - (6) define macro MAC5. (6) ends the macro definition, and causes the macro to be compiled.
- (7) - (9) are SAS code and are generated.
- (10) invokes macro MAC5. The macro processor executes (3) - (5), inserting the resulting code as if it had been typed between (9) and (11). Execution is similar to (4) - (6) in Example 4, as follows.
  - (2) defines a local macro variable.
  - (3) J is replaced with its value, 1. 1 is within the loop bounds, 1 to 3, so %DO loop processing continues.
  - (4) The following SAS code is generated: x1= y1
  - (5) %END ends the %DO statement, and the macro returns to (3).
  - (3) J is incremented from 1 to 2. 2 is within the loop bounds, 1 to 3, so %DO loop processing continues.
  - (4) The following SAS code is generated: x2= y2
  - (5) %END ends the %DO statement, and the macro returns to (3).
  - (3) J is incremented from 2 to 3. 3 is within the loop bounds, 1 to 3, so %DO loop processing continues.
  - (4) The following SAS code is generated: x3= y3
  - (5) %END ends the %DO statement, and the macro returns to (3).
  - (3) J is incremented from 3 to 4. 4 is outside the loop bounds, 1 to 3, so %DO loop processing ends, and processing resumes with (6).
(6) is a %MEND statement that ends the macro. At this point, the following SAS code has been generated, but not compiled or executed.

```sas
data two;
  set one;
  rename
x1 = y1
x2 = y2
x3 = y3
```

(11) and (12) are SAS code and are generated. (11) is a semicolon that completes the RENAME statement. (12), a RUN statement, is a step boundary that ends the DATA step and causes the following SAS statements to be compiled and executed.

```sas
data two;
  set one;
  rename
x1 = y1
x2 = y2
x3 = y3

run;
```

The SAS statements generated and then compiled and executed in this example are the same as in Example 4.

**Example 6. Use “Data driven” code to rename variables.**

The final example is more complex, and illustrates a more automated application.

We want to rename all variables in a data set that start with the letter “X” so that they are prefaced with “OLD_”. For example, rename “X” to “OLD_X” and “XYZ” to “OLD_XYZ”. We don’t know the number of variables that start with “X” or their names.

First, let’s construct a data set to work with. Three variables meet the selection criterion (start with “X”), and one does not.

```sas
data one;
x1=11;
y=22;
xxx=33;
xyz=44;
run;
```

Now, read the variable names that begin with "X" from a DICTIONARY table and copy them to macro variables. DICTIONARY tables are a set of read-only SAS data views that contain information about the current SAS session such as data libraries, SAS data sets, SAS macros, and external files in use or available and SAS system options settings. See the DICTIONARY Tables chapter in SAS 9.1.3 Language Reference: Concepts for an introduction to DICTIONARY tables, and The SQL Procedure chapter in the Base SAS 9.1.3 Procedures Guide for complete information. For a more detailed explanation of similar examples, see Gilsen (2008).

```sas
proc sql noprint;                      (1)
  select name
  into :var1-:var&sysmaxlong
  from dictionary.columns
  where libname="WORK"
  and memname="ONE"
  and upcase(substr(name,1,1))="X";
%let num_vars=&sqllobs;
quit;
```

This code executes as follows.

- The first variable name that meets the selection criteria (starts with upper case “X” in data set WORK.ONE) is copied to macro variable VAR1, the second variable name that meets the selection criteria is copied to macro variable VAR2, and so on.
SYSMAXLONG is an operating system-specific automatic macro variable containing the largest possible integer value. Specifying SYSMAXLONG as the upper bound of the range of macro variables allows the number of macro variables created by PROC SQL to be data driven. In this example, three variables meet the selection criteria, so macro variables VAR1, VAR2, and VAR3 are created with the following values.

<table>
<thead>
<tr>
<th>Macro variable</th>
<th>Value</th>
</tr>
</thead>
<tbody>
<tr>
<td>VAR1</td>
<td>x1</td>
</tr>
<tr>
<td>VAR2</td>
<td>xxx</td>
</tr>
<tr>
<td>VAR3</td>
<td>xyz</td>
</tr>
</tbody>
</table>

The automatic macro variable SQLOBS contains the number of rows selected by the last PROC SQL statement. In this example, the value of SQLOBS is 3.

Now, execute PROC DATASETS and use a RENAME statement to rename the variables. For each variable name, macro MAC6 generates the text `varname=OLD_varname`, which is used as a clause in the RENAME statement.

```%macro mac6;                         (2) %local j;                          (3) %do j = 1 %to &num_vars;           (4) &&var&j = old_&&var&j            (5) %end;                              (6) %mend mac6;                          (7) proc datasets library=work nolist;   (8) modify one;                        (9) rename                             (10) %mac6                            (11) ;                                  (12) run; quit;                           (13)```

Before explaining this program, let's briefly focus on the double ampersand (&&) in (5). The double ampersand (&&) is called an "indirect reference" to a macro variable. The macro processor resolves two ampersands to one ampersand (&& to &) rather than applying the & to the text that follows.

The first time the loop executes, macro variable J is 1, and the macro processor processes `&&VAR&J` left to right in two passes as follows:

**Pass 1:**
1. && resolves to &
2. VAR is text and is unchanged
3. &J is resolved to 1
**Result:** `&&VAR&J` is resolved to `&VAR1`

**Pass 2:**
1. `&VAR1` is resolved to x1, so `&&VAR&J` is replaced by x1 in two places, generating `x1=old_x1`

The second time the loop executes, macro variable J is 2, and the macro processor processes `&&VAR&J` left to right in two passes as follows:

**Pass 1:**
1. && resolves to &
2. VAR is text and is unchanged
3. &J is resolved to 2
**Result:** `&&VAR&J` is resolved to `&VAR2`

**Pass 2:**
1. `&VAR2` is resolved to xxx, so `&&VAR&J` is replaced by xxx in two places, generating `xxx=old_xxx`

The && is needed to make the macro processor scan twice. If you code `&VAR&J` instead of `&&VAR&J`, the macro processor tries to resolve &VAR and &J in a single pass and concatenate the values. This generates an error unless &VAR was previously defined (in that case, there is no error message but an unexpected result).

Now, let's review this program, which executes similarly to Example 5.
(2) - (7) define macro MAC6. (7) ends the macro definition, and causes the macro to be compiled.

(8) - (10) are SAS code and are generated.

(11) invokes macro MAC6. The macro processor executes statements as follows.

(3) defines a local macro variable.

(4) J is replaced with its value, 1. 1 is within the loop bounds, 1 to 3, so %DO loop processing continues.
(5) The following SAS code is generated: x1 = old_x1
(6) %END ends the %DO statement, and the macro returns to (6).

(4) J is incremented from 1 to 2. 2 is within the loop bounds, 1 to 3, so %DO loop processing continues.
(5) The following SAS code is generated: xxx = old_xxx
(6) %END ends the %DO statement, and the macro returns to (6).

(4) J is incremented from 2 to 3. 3 is within the loop bounds, 1 to 3, so %DO loop processing continues.
(5) The following SAS code is generated: xyz = old_xyz
(6) %END ends the %DO statement, and the macro returns to (6).

(4) J is incremented from 3 to 4. 4 is outside the loop bounds, 1 to 3, so %DO loop processing ends, and macro processing resumes with (7).

(7) is a %MEND statement that ends the macro. At this point, the following SAS code has been generated, but not compiled or executed.

```sas
proc datasets library=work nolist;
  modify one;
  rename
    x1 = old_x1
    xxx = old_xxx
    xyz = old_xyz
```

(12) and (13) are SAS code and are generated. (12) is a semicolon that completes the RENAME statement, and (13) is a step boundary that ends PROC DATASETS and causes the following SAS statements to be compiled and executed.

```sas
proc datasets library=work nolist;
  modify one;
  rename
    x1 = old_x1
    xxx = old_xxx
    xyz = old_xyz
;
  run; quit;
```

Note that a more robust version of this example would include the following tests.

- That no existing variable name is longer than 28 characters, since OLD_ adds 4 characters to the name and the maximum variable name length is 32 characters.
- That there are no cases where variable names OLD_ Xname and Xname both exist. That is, if the data set has variables XYZ and OLD_XYZ, the RENAME statement generates an error when it tries to rename XYZ to OLD_XYZ.

An alternate way to code this example is to copy the variables that meet the selection criterion to a single space-separated macro variable, use the automatic macro variable SQLOBS to determine the number of variable names in data set WORK.ONE that begin with "X", and use the %SCAN function to process the variable names, as follows. The PROC DATASETS step and the results are the same as above.
proc sql noprint;
   select name into :var_names separated by ' ' 
   from dictionary.columns 
   where libname="WORK" 
   and memname="ONE" 
   and upcase (substr(name,1,1))="X";
%let num_values=&sqlobs;
quit;

%macro mac6;
   %local j;
   %do j = 1 %to &num_values;
      %scan(&var_names, &j) = old_%scan(&var_names, &j)
   %end;
%mend mac6;

proc datasets library=work nolist;
   modify one;
   rename
      %mac6
   ;
   run; quit;

CONCLUSION

This paper explained in a simple, informal way the interaction between SAS DATA and PROC step code and macros. Understanding this interaction allows SAS users to understand existing applications and code their own applications more effectively.

For more information, contact the author, Bruce Gilsen, by mail at Federal Reserve Board, Mail Stop 157, Washington, DC 20551; by e-mail at bruce.gilsen@frb.gov; or by phone at 202-452-2494.

REFERENCES


ACKNOWLEDGMENTS

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

The following people contributed extensively to the development of this paper: Donna Hill, Bill Jackson, and Heidi Markovitz at the Federal Reserve Board, and Ian Whitlock. Their support is greatly appreciated.